
You can distribute the following freely for non-commercial use provided you retain the credit to me and periodically send me appreciative e-mails.
READER BEWARE - undergrads should read with caution - sometimes the advice re writing and analysis here contradicts what is advised in your courses.  Obviously you must follow the advice given in courses.  The discrepancies could be because 1) undergrad stats is an idealized version of reality whereas postgrads grapple with real data and publication pressure 2) statistical decision-making requires personal choices and different profs & practitioners may differ.
A wise practice for any write-up is to scan the intended publication outlet (the journal, other theses in the same lab, etc.) and try to find 3 or 4 examples of how the analysis has been written up before there to serve as models.

Writing up higher-order interactions in ANOVA

A write-up for a two-way interaction in ANOVA has two parts: the text and a graph.  The graph shows the simple effects of the IV at high and low levels of the moderator.  Because the IVs in ANOVA are categorical moderator, technically you should have bar graphs, not line graphs.  The write-up begins with an overview paragraph under the heading ‘design’ or ‘overview’ describing the analysis and treatment of missing variables and outliers, along with how people are dealing with the sphericity assumption in mixed or within ANOVA.  Then there are often separate sections of the results for each DV.  Within the sections typically people describe the main effects in terms of F, df, p, and eta2 or partial eta2 and then the interaction with the same statistics.  The simple effects are then reported and the reader is referred to Figure 1 (i.e., the graph of the interaction).
For higher-order interactions, you can report the main effects and build up, dealing with the simple effects and simple interactions / simple simple effects as you go.  However, this makes a very wordy presentation.  If the higher-order interaction is significant and theoretically expected, I would normally not report any lower order omnibus effects, but only the highest interaction and its simple effects.  “As expected, the four-way interaction was significant, F()=, p=, eta2=.”  However, reviewers can be cranky either way.
You report F, p, and effect size measure (normally partial eta2, because it’s easy to get out of SPSS) with every effect, and you report the means with the main effects and simple effects.  NB if you have more than 2 means for an IV, you also need to do follow-up tests to reveal where among the 3+ means the significant difference(s) is/are found.

Often times with more complex designs the graph is eliminated because too many are required to show the interaction and graphs are expensive, in manuscripts.  In theses, you may end up with multiple graphs – e.g., you could decompose a 4-way by showing 4 simple simple interactions (each graph showing the simple simple simple effect of variable A for B1 and B2, and the graphs being for C1D1, C1D2, C2D1, and C2D2.  In both cases, a Table of the cell means and SDs is normally included.
In decomposing a higher-order interaction (e.g., a 4-way in which each IV has only 2 levels), technically you have numerous lower-order interactions (4 3-ways and 6 2-ways) and 8 simple simple simple effects to look at.  Writing them all up is mind-numbing for the reader and it’s important to simplify as much as possible.
As a general strategy, I prefer to go straight to the simple effects of the most important IV (chosen on theoretical grounds).  For example, I would go straight from the 4-way to the simple simple simple effects of the most theoretically important variable, with an accompanying table of means and SDs.

If you are conducting exploratory analyses or if you are predicting that some simple lower-order interactions are significant while others aren’t, you can do the 4-way followed by the simple 3-ways, the simple simple 2-ways, and the simple simple simple effects.  You still need to think about which moderators you use in which order because it is frowned on to report multiple versions of the simple effects because of the increased experimentwise error.

Often times in the literature people will report a significant higher order interaction and then let’s say 1 simple lower-order interaction that’s sig and another that’s ns.  Then they will stop following up the ns interaction and only follow up the sig simple lower order interaction.  NB this is common and accepted but also means you’re not reporting what actually is going on for the data where the lower order interaction is ns (e.g., if a simple 3-way is ns, does that mean there’s no differences in the data or that there are 2-way interactions or main effects that are unqualified?). 
How to do this in SPSS

1. Look at the variables.  Consider intercorrelations among measured IVs especially.
2. Run the omnibus tests.  Choose IV and moderator(s).

3. Test simple interactions in SPSS (optional)
4. Test simple effects.

1. Use analyse > descriptive > frequencies to get descriptive statistics and histograms for the data.  Have a look for errors and violations of assumptions.  Never skip this step.  Note the uncentered means and standard deviations here are informative but don’t have listwise deletion.  
FREQUENCIES

  VARIABLES=iv1 iv2 iv3 iv4 dv1 dv2 

  /STATISTICS=STDDEV MINIMUM MAXIMUM SEMEAN MEAN MEDIAN SKEWNESS  SESKEW

  KURTOSIS SEKURT

  /HISTOGRAM

  /ORDER=  ANALYSIS .

Run this syntax.  Correct any errors such as out-of-range data.

If your IVs are experimental and you have equal n, they will be uncorrelated.  However, if you have a quasi-experimental design with measured variables like gender or nationality unequal n is extremely common.

1. Use Analyze > Correlate > Bivariate

2. enter all ivs and DVs

3. click options > “Exclude cases listwise” and in the same window “Means and standard deviations” > continue

4. click paste

CORRELATIONS

  /VARIABLES= iv1 iv2 iv3 iv4 dv1 dv2 

  /PRINT=TWOTAIL NOSIG

  /STATISTICS DESCRIPTIVES

  /MISSING=LISTWISE .

A rule of thumb is that if any 2 ivs are correlated more than .3, you may want to drop one of them or combine them.  Leaving correlated IVs as “independent” predictors can lower power but also introduce instability so you get significant results that don’t replicate (and possibly make no sense).   Instability of results for interactions is exacerbated by correlated IVs.
********************************************************************

*SYNTAX FOR a four-way between groups design
********************************************************************

Analyse > General Linear Model > Univariate ; 
· enter DV and IVs ; 
· click on options.  Using the mouse, highlight all the IVs and their interactions in the “Factors and factor interaction(s)” and use the arrow to copy them into the the “Display means for” box.  Tick on “Compare main effects”.
· Still in the options, tick on Display Descriptive statistics, estimates of effect size, observed power, and homogeneity tests.

· If you have more than 2 levels of any factor, go to post hoc and pick one (e.g., Games-Howell Tukey is a good post hoc test that can deal with heterogeneous variance).

· Hit paste.

UNIANOVA

  chooseig BY strength salience id_hl sali_hl

  /RANDOM = strength salience id_hl sali_hl

  /METHOD = SSTYPE(3)

  /INTERCEPT = INCLUDE

  /EMMEANS = TABLES(OVERALL)

  /EMMEANS = TABLES(strength) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(salience) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(id_hl) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(sali_hl) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(strength*salience)

  /EMMEANS = TABLES(strength*id_hl)

  /EMMEANS = TABLES(salience*id_hl)

  /EMMEANS = TABLES(strength*salience*id_hl)

  /EMMEANS = TABLES(strength*sali_hl)

  /EMMEANS = TABLES(salience*sali_hl)

  /EMMEANS = TABLES(strength*salience*sali_hl)

  /EMMEANS = TABLES(id_hl*sali_hl)

  /EMMEANS = TABLES(strength*id_hl*sali_hl)

  /EMMEANS = TABLES(salience*id_hl*sali_hl)

  /EMMEANS = TABLES(strength*salience*id_hl*sali_hl)

  /PRINT = DESCRIPTIVE ETASQ OPOWER HOMOGENEITY

  /CRITERIA = ALPHA(.05)

  /DESIGN = strength salience id_hl sali_hl strength*salience strength*id_hl

  salience*id_hl strength*salience*id_hl strength*sali_hl salience*sali_hl

  strength*salience*sali_hl id_hl*sali_hl strength*id_hl*sali_hl salience*id_hl*sali_hl

  strength*salience*id_hl*sali_hl .

*Put the cursor in the syntax and hit control r or the little arrow in the syntax window to run the syntax.  You report the omnibus tests from this main analysis (“Tests of between-subjects effects” table contains F, p value, and partial eta2).  And you should pause to look at the descriptive stats table and the Levine’s test to see if you have extremely unequal n (some cells 3 x as big as other cells) or heterogenous variance.  If so, it can result in lower power and/or instability.  But normally no one reports this.  If you have any significant main effects, you can look at the tables of estimated marginal means for the means and pairwise comparisons.  The latter can be reported if there are more than 3 means, but for manuscripts its better to use post hoc tests or planned comparisons because of the increased alpha.

If you are able to justify, on theoretical grounds, going straight to the simple simple simple effects, your life is extremely easy.  On the line in the syntax for the 4-way interaction marginal means table simply add the phrase “compare(ivname)” referring to the IV in which you are interested.  If you then go to the section in the estimated marginal means for the 4-way, you will find the cell means and (in the ‘univariate tests’ section) tests of the simple simple simple effects of key IV.


For example, using the syntax below will output tests of the simple simple simple effects of the variable SALIHL for each combination of levels of the other three IVs.  

UNIANOVA

  chooseig BY strength salience id_hl sali_hl

  /RANDOM = strength salience id_hl sali_hl

  /METHOD = SSTYPE(3)

  /INTERCEPT = INCLUDE

  /EMMEANS = TABLES(OVERALL)

  /EMMEANS = TABLES(strength) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(salience) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(id_hl) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(sali_hl) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(strength*salience)

  /EMMEANS = TABLES(strength*id_hl)

  /EMMEANS = TABLES(salience*id_hl)

  /EMMEANS = TABLES(strength*salience*id_hl)

  /EMMEANS = TABLES(strength*sali_hl)

  /EMMEANS = TABLES(salience*sali_hl)

  /EMMEANS = TABLES(strength*salience*sali_hl)

  /EMMEANS = TABLES(id_hl*sali_hl)

  /EMMEANS = TABLES(strength*id_hl*sali_hl)

  /EMMEANS = TABLES(salience*id_hl*sali_hl)

  /EMMEANS = TABLES(strength*salience*id_hl*sali_hl) compare(sali_hl)
  /PRINT = DESCRIPTIVE ETASQ OPOWER HOMOGENEITY

  /CRITERIA = ALPHA(.05)

  /DESIGN = strength salience id_hl sali_hl strength*salience strength*id_hl

  salience*id_hl strength*salience*id_hl strength*sali_hl salience*sali_hl

  strength*salience*sali_hl id_hl*sali_hl strength*id_hl*sali_hl salience*id_hl*sali_hl

  strength*salience*id_hl*sali_hl .

You would then put all the simple effects in a table and report them in simplified form in the text.  E.g., with one imaginary set of results we might report “Simple effect tests for sali_hl at each combination of levels of the other factors revealed that for high identifiers in high power groups in the group salience condition, high perceived salience (M=3.00) resulted in higher intentions to act than low perceived salience (M=4.00) , F()=, p=, partial eta2=, but in the other conditions the differences were not significant, Fs < 1.39, ps > .116, partial eta2s < .02.”  If it gets much more complex than that, you might simply describe the table without any #s.  “Simple effect tests for sali_hl at each combination of levels of the other factors are reported in Table 1.  As may be seen in Table 1, the effect of Sali_HL was significantly positive at high strength, regardless of identity and salience, and significantly negative at low strength for low identifiers in the personal identity salience condition.  However, there were no effects in the other conditions.”
Graphs would be unusual in manuscripts but are good for theses.  For a 4-way you would probably show four simple simple interactions (or perhaps only the significant simple simple interactions), choosing the graph moderators based on theory.
NB that the simple effect tests in the syntax above use a pooled error term, which maximises your power.  However, if you have heterogeneous variance, it is also permissible to split the file and look at the results for each group.  To do this, go to “DATA > SPLIT FILE” and click “compare groups”.  Select your three moderators and put them into the “Groups Based on:” window.  Hit paste.  No go to “ANALYSES > GLM > UNIVARIATE” and take the moderating variables out of the IV list.  Hit paste.  You should end up with syntax like this:

SORT CASES BY strength salience id_hl  .

SPLIT FILE

  LAYERED BY strength salience id_hl .

UNIANOVA

  chooseig BY sali_hl

  /RANDOM = sali_hl

  /METHOD = SSTYPE(3)

  /INTERCEPT = INCLUDE

  /EMMEANS = TABLES(OVERALL)

  /EMMEANS = TABLES(sali_hl) COMPARE ADJ(LSD)

  /PRINT = DESCRIPTIVE ETASQ OPOWER HOMOGENEITY

  /CRITERIA = ALPHA(.05)

  /DESIGN = sali_hl .

*Type “SPLIT FILE OFF.” Below the last line, not forgetting the period.  Highlight and run the whole syntax, from sort to off. 
SORT CASES BY strength salience id_hl  .

SPLIT FILE

  LAYERED BY strength salience id_hl .

UNIANOVA

  chooseig BY sali_hl

  /RANDOM = sali_hl

  /METHOD = SSTYPE(3)

  /INTERCEPT = INCLUDE

  /EMMEANS = TABLES(OVERALL)

  /EMMEANS = TABLES(sali_hl) COMPARE ADJ(LSD)

  /PRINT = DESCRIPTIVE ETASQ OPOWER HOMOGENEITY

  /CRITERIA = ALPHA(.05)

  /DESIGN = sali_hl .

SPLIT FILE OFF .
*You will notice that the n are incredibly small, reflecting the fact that with normal social psych samples this is a losing strategy (power too low, so that no matter now much heterogeneity of variance it’s not good to do). 
If you take the treatment mean square from the simple effects from each combo of the moderators and divide by the error term from the original 4-way interaction you have the appropriate simple effect test using the pooled error term.  However, it’s easier to use the compare command in EMMEANS described above.
What if you want to go through the whole shebang, from the 4-way to the simple 3-ways to the simple simple 2ways, before you get to the simple simple simple effects?  There are two options.  You can split the file as demonstrated above except only split on one moderator (chosen on theoretical ground) and only delete one from the syntax:
SORT CASES BY id_hl  .

SPLIT FILE

  LAYERED BY id_hl .
UNIANOVA

  chooseig BY sali_hl strength salience

  /RANDOM = sali_hl strength salience

  /METHOD = SSTYPE(3)

  /INTERCEPT = INCLUDE

  /EMMEANS = TABLES(OVERALL)

  /EMMEANS = TABLES(sali_hl) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(strength) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(salience) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(sali_hl*strength)

  /EMMEANS = TABLES(sali_hl*salience)

  /EMMEANS = TABLES(strength*salience)

  /EMMEANS = TABLES(sali_hl*strength*salience)

  /PRINT = DESCRIPTIVE ETASQ OPOWER HOMOGENEITY

  /CRITERIA = ALPHA(.05)

  /DESIGN = sali_hl strength salience sali_hl*strength sali_hl*salience strength*salience

  sali_hl*strength*salience .

SPLIT FILE OFF .

This will give you separate error term tests of the 3-ways (justifiable if you have heterogeneous variance, but often lower power).
If you take the treatment mean squares from each of the three-way interactions at high and low ID and divide them by the error term from the original F, you have an appropriate pooled error term test.  This delivers better power.
Some people think you can get simple interactions out of the MANOVA command.  (First you have to rescore your IVs – MANOVA will only accept positive sequential integers e.g., 1, 2, 3). Then you could use syntax, e.g., from http://www.coe.unt.edu/brookshire/spss3way.htm#simpintr .
In the first line, SCORE is the DV name and the parentheses after all the IV names A B and C define the lowest and highest level of the IV (here 3 IVs with 3 levels).

* Simple Interaction Effects for A BY B at C(l).

* All termsnot involving A  or A BY B are included in the DESIGN statement.

*    before the simple interaction effect statements.

MANOVA SCORE BY A(1,3) B(1,3) C(1,3)

      /DESIGN= A, B, C, A BY B, A BY C, B BY C, A BY B BY C

      /DESIGN= C, B BY C, A BY B WITHIN C(1), A BY B WITHIN C(2)

               A BY B WITHIN C (3)

      /DESIGN= B,C, B BY C, A BY C WITHIN B(1),A BY C WITHIN B(2),

               A BY C WITHIN B(3)

      /DESIGN= A,C, A BY C, B BY C WITHIN A(1) B BY C WITHIN A(2)

               B BY C WITHIN A(3).

I would strongly advise *NOT* to use the within command in MANOVA or the mwithin command in MANOVA.  MANOVA adjusts the treatment effect based on the other variables in the design and wsdesign lines (“corrected” means) in a manner that I have found difficult to predict or understand. (E.g., if you compare the treatment SS from the MANOVA within to the treatment SS if you split the file you may find an alarming difference. Or if you add and delete other treatment effects from the design.)
If you use MANOVA it is essential to understand that the other terms in the design line change the significance profoundly and if you do not understand the logic – and I do not – you’re quite likely to make Type I or Type II errors.  (NB, if you only put in the interactions that you are looking for, you get output but the wrong output (wrong treatment SS/df, wrong p value) – this stuff is too dangerous to use IMHO).
So for simple interactions, either use separate error terms or use the treatment MS from the separated analyses divided by the error MS from the original omnibus interaction.  If you have an aversion to hand calculations you should learn to use L-matrix (bs contrasts), M-matrix (ws contrasts and mixed) and/or special contrast.
This link looks helpful: http://www.nicholasgibson.com/lmatrix.html .
********************************************************************

*SYNTAX FOR a four-way within subjects (repeated measures) design

********************************************************************

Analyse > General Linear Model > Repeated Measures ; 

· enter names of WS factors and the # of levels of each, hit ok;

· fill in the dvs as appropriate;

· click on options.  Using the mouse, highlight all the IVs and their interactions in the “Factors and factor interaction(s)” and use the arrow to copy them into the the “Display means for” box.  Tick on “Compare main effects”.

· Still in the options, tick on Display Descriptive statistics, estimates of effect size, observed power, and homogeneity tests.

· If you have more than 2 levels of any factor, go to contrasts and change them from polynomial to one that makes sense (e.g., repeated, difference, simple).

· Hit paste.

GLM

  opa4 opb5 opc6 opd7 opa8 opb9 opc10 opd11 opa12 opb13 opc14 opd15 opa16

  opb17 opc18 opd19

  /WSFACTOR = factor1 2 Polynomial B 2 Polynomial C 2 Polynomial D 2

  Polynomial

  /METHOD = SSTYPE(3)

  /EMMEANS = TABLES(OVERALL)

  /EMMEANS = TABLES(factor1) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(B) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(C) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(D) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(factor1*B)

  /EMMEANS = TABLES(factor1*C)

  /EMMEANS = TABLES(B*C)

  /EMMEANS = TABLES(factor1*B*C)

  /EMMEANS = TABLES(factor1*D)

  /EMMEANS = TABLES(B*D)

  /EMMEANS = TABLES(factor1*B*D)

  /EMMEANS = TABLES(C*D)

  /EMMEANS = TABLES(factor1*C*D)

  /EMMEANS = TABLES(B*C*D)

  /EMMEANS = TABLES(factor1*B*C*D)

  /PRINT = DESCRIPTIVE ETASQ OPOWER

  /CRITERIA = ALPHA(.05)

  /WSDESIGN = factor1 B C D factor1*B factor1*C B*C factor1*B*C factor1*D B*D

  factor1*B*D C*D factor1*C*D B*C*D factor1*B*C*D .

You report the omnibus tests from this main analysis (“Tests of within-subjects effects” table contains F, p value, and partial eta2).  Because there are WS factors, you need to specify which p value you reported.  (“In the analyses below, the Greenhouse-Geisser correction was employed but the full degrees of freedom are reported.”)

Again, you should pause to look at the within-subjects factors table (did you define the repeated measures correctly?), the descriptive stats table (do the means look ok)? Ignore the Mauchly’s test and the multivariate tests, as well as the ws contrasts (unless you asked for something). 

Assuming an absence of errors, proceed as above.  Given a significant interaction, it’s best to go straight to the simple effects using COMPARE syntax, if this can be justified.

If you need to report the simple interaction(s), you can again get the treatment effects by editing the DVs and wsfactor line so that you are looking at the cell means you want and dividing by the original pooled error term.  However, in WS ANOVA it is even more appropriate to use separate error terms (you can cite Howell on this, among others, but it is common practice) and there is generally less of a tradeoff of power.

*Simple threeway interaction at D1.

GLM

  opa4 opb6 opc8 opd10 opa12 opb14 opc16 opd18
  /WSFACTOR = factor1 2 Polynomial B 2 Polynomial C 2 

  Polynomial

  /METHOD = SSTYPE(3)

  /EMMEANS = TABLES(OVERALL)

  /EMMEANS = TABLES(factor1) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(B) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(C) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(factor1*B)

  /EMMEANS = TABLES(factor1*C)

  /EMMEANS = TABLES(B*C)

  /EMMEANS = TABLES(factor1*B*C)

  /PRINT = DESCRIPTIVE ETASQ OPOWER

  /CRITERIA = ALPHA(.05)

  /WSDESIGN = factor1 B C factor1*B factor1*C B*C factor1*B*C .

*Simple three-way at D2.

GLM

  Opa5 opb7 opc9 opd11 opa13 opb15 opc17 opd19
  /WSFACTOR = factor1 2 Polynomial B 2 Polynomial C 2 

  Polynomial

  /METHOD = SSTYPE(3)

  /EMMEANS = TABLES(OVERALL)

  /EMMEANS = TABLES(factor1) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(B) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(C) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(factor1*B)

  /EMMEANS = TABLES(factor1*C)

  /EMMEANS = TABLES(B*C)

  /EMMEANS = TABLES(factor1*B*C)

  /PRINT = DESCRIPTIVE ETASQ OPOWER

  /CRITERIA = ALPHA(.05)

  /WSDESIGN = factor1 B C factor1*B factor1*C B*C factor1*B*C .

Do *NOT* use the within command in MANOVA or the mwithin command in MANOVA to calculate simple interactions – this adjusts the treatment effect based on the other variables in the design and wsdesign lines in a manner that I have found difficult to predict or understand.  For simple interactions in a WS design, use separate error terms or learn to use L-matrix (bs contrasts) M-matrix (ws contrasts and mixed) or special contrast.
This link looks helpful: http://www.nicholasgibson.com/lmatrix.html .

********************************************************************

*SYNTAX FOR a four-way mixed design

********************************************************************

Analyse > General Linear Model > Repeated Measures ; 

· enter names of WS factors and the # of levels of each, hit ok;

· fill in the dvs as appropriate;

· Add in the bs IVs 

· click on options.  Using the mouse, highlight all the IVs and their interactions in the “Factors and factor interaction(s)” and use the arrow to copy them into the the “Display means for” box.  Tick on “Compare main effects”.

· Still in the options, tick on Display Descriptive statistics, estimates of effect size, observed power, and homogeneity tests.

· If you have more than 2 levels of any WS factor, go to contrasts and change them from polynomial to one that makes sense (e.g., repeated, difference, simple).  If you have more than 2 levels of any BS factor, pick a post hoc test that makes sense (e.g., Games-Howell)

· Hit paste.

GLM

  opa4 opb5 opc6 opd7 BY strength salience

  /WSFACTOR = factor1 2 Polynomial B 2 Polynomial

  /METHOD = SSTYPE(3)

  /EMMEANS = TABLES(OVERALL)

  /EMMEANS = TABLES(factor1) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(B) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(factor1*B)

  /EMMEANS = TABLES(strength) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(salience) COMPARE ADJ(LSD)

  /EMMEANS = TABLES(strength*salience)

  /EMMEANS = TABLES(strength*factor1)

  /EMMEANS = TABLES(salience*factor1)

  /EMMEANS = TABLES(strength*salience*factor1)

  /EMMEANS = TABLES(strength*B)

  /EMMEANS = TABLES(salience*B)

  /EMMEANS = TABLES(strength*salience*B)

  /EMMEANS = TABLES(strength*factor1*B)

  /EMMEANS = TABLES(salience*factor1*B)

  /EMMEANS = TABLES(strength*salience*factor1*B)

  /PRINT = DESCRIPTIVE ETASQ OPOWER

  /CRITERIA = ALPHA(.05)

  /WSDESIGN = factor1 B factor1*B

  /DESIGN = strength salience strength*salience .

You report the omnibus tests from this main analysis (“Tests of within-subjects effects” for the WS effects and the mixed interactions, and “tests of between-subjects effects” for the BS alone.  The tables contain F, p value, and partial eta2 which are to be reported.  Because there are WS factors, you need to specify which p value you reported.  (“In the analyses below, the Greenhouse-Geisser correction was employed for the analyses involving repeated-measures factors, but the full degrees of freedom are reported.”)

Again, you should pause to look at the within-subjects factors table (did you define the repeated measures correctly?), the descriptive stats table (do the means look ok) and the Levine test (how’s the heterogeneity)? 

Again, you should ignore the Mauchly’s test and the multivariate tests, as well as the ws contrasts (unless you asked for something specific). 

Assuming an absence of errors, proceed as above.  Given a significant interaction, it’s best to go straight to the lowest possible simple effects using COMPARE syntax, if this can be justified.

If you need to report the simple interaction(s) within a repeated measures factor, you can again get the treatment effects by editing the DVs and wsfactor line so that you are looking at the cell means you want.  If you are looking within a BS factor, you again can get the simple interactions by splitting the file on the BS effect.

In each case, you can justify using the separate error term but for analyses within the BS factor it’s likely to drop your power harshly and is less justifiable.  Dividing by the original pooled error term from the interaction test will give you a stronger test.  For the WS ANOVA it is even more appropriate to use separate error terms (you can cite Howell on this, among others, but it is common practice) and there is generally less of a tradeoff of power.

Do *NOT* use the within command in MANOVA or the mwithin command in MANOVA for simple interactions – this adjusts the treatment effect based on the other variables in the design and wsdesign lines in a manner that I have found difficult to predict or understand. 
Either use separate error term analyses, or use the treatment MS from the new separate analysis divided by the error MS from the original omnibus interaction.  If you don’t like hand calculations, you should learn to use L-matrix (bs contrasts), M-matrix (ws contrasts and mixed) and/or special contrast.  This link looks helpful: http://www.nicholasgibson.com/lmatrix.html .

In addition, there is a page here which has syntax for a mixed simple interaction:

http://listserv.uga.edu/cgi-bin/wa?A2=ind0107&L=spssx-l&P=13210 
GLM w1 w2 w3 BY a b

   /WSFACTOR w 3 Polynomial

   /LMATRIX="interaction A*W at B1"

        A  1 -1  0   A*B  1  0   -1  0    0  0;

        A  1  0 -1   A*B  1  0    0  0   -1  0

   /LMATRIX="interaction A*W at B2"

        A  1 -1  0   A*B  0  1    0 -1   0  0;

        A  1  0 -1   A*B  0  1    0  0   0 -1

   /MMATRIX "pooled W difference"

        all  1 -1  0;

        all  1  0 -1

   /WSDESIGN w

   /DESIGN a b a*b .

But NB if you have other WS factors or BS factors which are not being included, the syntax changes again – read carefully.  The matrix language in SPSS is difficult but better than the MANOVA command which is opaque (you can’t easily tell what it has done to the means) and easily outputs misleading stats.
